

Supplemental Digital Content 1:

SAS code accompanying “Confidence intervals for the interaction contrast ratio: A simple solution with SAS PROC NLMIXED and SAS PROC NLP”

Code 1: Reading the example data set and fitting the ordinary logistic regression model with PROC LOGISTIC.

```
DATA rifa;
    INPUT protect iriscolor ncases nobs;
    CARDS;
    0 0 50 195
    0 1 309 833
    1 0 12 53
    1 1 84 195
;RUN;

PROC LOGISTIC DATA=rifa DESCENDING;
    MODEL ncases/nobs=protect iriscolor protect*iriscolor;
RUN;
```

Code 2: Fitting the ordinary logistic regression model with PROC NLMIXED and calculating the ICR and its confidence interval with the ESTIMATE statement. Using PROC NLMIXED we are forced to write down explicitly (1) the parameters to be estimated for the original logistic model with starting values (via the PARMS statement), and (2) the model equation (via the definition of p, the event probability, and the MODEL statement). Note that we set df=10000 in the PROC NLMIXED statement which gives standard Wald confidence intervals (because the t-distribution with 10.000 degrees of freedom essentially equals the standard normal distribution).

```
PROC NLMIXED DATA=rifa DF=10000;
  PARMS beta0=-1.06 b_protect=-0.164 b_iriscolor=0.537
        b_interaction=0.413;

  p=    exp(beta0+b_protect*protect+b_iriscolor*iriscolor+
          b_interaction*protect*iriscolor)/
        (1+ exp(beta0+b_protect*protect+b_iriscolor*iriscolor+
          b_interaction*protect*iriscolor));

  MODEL ncases~BINOMIAL(nobs,p);

  ESTIMATE "ICR" exp(b_protect+b_iriscolor+b_interaction)-
              exp(b_protect)-exp(b_iriscolor)+1;

RUN;
```

Code 3: Fitting the reparametrized logistic regression model with PROC NLP and calculating confidence intervals for the ICR with the PROFILE statement. Similar to code 2 we are forced to give starting values for the parameters via the PARMS statement (in PROC NLP separated by commas) and to write down the model equation and the likelihood function explicitly. The MAX statement declares that the specified likelihood function should be maximized. Additionally, the VARDEF=N option in the PROC NLP statement ensures the correct computation of the covariance matrix of the parameters. Equation (1) is specified as a simple program statement in the procedure code. Note that the definition of the event probability p still uses the $b_{\text{interaction}}$ term. The crucial statement for estimating a PL confidence interval is the PROFILE statement which names the parameters (here the ICR) for which these intervals should be computed. Additionally the desired alpha level can be specified.

```
PROC NLP DATA=rifa VARDEF=N;
  PARMS beta0=-1.06, b_protect=-0.164, b_iriscolor=0.537,
        ICR=0.636;
  b_interaction=log((ICR+exp(b_protect)+
                    exp(b_iriscolor)-1)/
                    (exp(b_protect)*exp(b_iriscolor)));
  p= exp(beta0+b_protect*protect+b_iriscolor*iriscolor+
        b_interaction*protect*iriscolor)/
      (1+ exp(beta0+b_protect*protect+b_iriscolor*iriscolor+
              b_interaction*protect*iriscolor));
  loglike=(ncases*log(p) + (nobs-ncases)*log(1-p));
  MAX loglike;
  PROFILE ICR / ALPHA=0.05;
RUN;
```